# Maintainer: Balló György <ballogyor+arch at gmail dot com>
# Contributor: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Contributor: Andrea Scarpino <andrea@archlinux.org>
# Contributor: Javier 'Phrodo_00' Aravena <phrodo.00@gmail.com>
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>

pkgname=devede
_pkgname=devedeng
pkgver=4.17.0
pkgrel=2
pkgdesc='Program to create VideoDVDs and CDs'
arch=('any')
url='https://rastersoft.com/programas/devede.html'
license=('GPL3')
depends=('gtk3' 'python-cairo' 'python-gobject' 'python-setuptools' 'ffmpeg' 'cdrtools' 'dvdauthor'
         'vcdimager')
source=("https://gitlab.com/rastersoft/$_pkgname/-/archive/$pkgver/$_pkgname-$pkgver.tar.gz")
sha256sums=('ef50effc72404c6cc4e87244ab8f622179e82368da422756d70316feb94de032')

prepare() {
  cd $_pkgname-$pkgver
  sed -i "/share', 'pixmaps/d" setup.py
}

build() {
  cd $_pkgname-$pkgver
  python setup.py build
}

package() {
  cd $_pkgname-$pkgver
  python setup.py install --root="$pkgdir" --optimize=1 --skip-build
}
